HRP 262, SAS LAB ONE, April 15, 2009

Lab One: 

Introduction to time-and-date formats, time-to-event variables, Kaplan-Meier curves, plotting, parametric regression (if time)
Lab Objectives:

After today’s lab you should be able to:

1. Manipulate and format date and time variables in SAS.

2. Use PROC FORMAT to create user-defined formats.

3. Put data into the correct structure for survival analysis: create time and censor variables. 

4. Quickly examine univariate distributions and identify outliers via point-and-click features.

5. Produce enhanced graphs using PROC GPLOT.  

a. Use the TITLE, SYMBOL, and AXIS statements (which are global statements).

b. Use different symbols for different values of a classification variable (such as censored/failed).

c. Export graphs as image files.

d. Know where to go for help on SAS/GRAPH: http://v8doc.sas.com/sashtml/
6. Produce a simple Kaplan-Meier curve (we will continue this in lab next week).

7. Use PROC LIFEREG to carry out a simple parametric (exponential) regression and interpret the results (we will continue this in lab next week).

LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Save the excel dataset “hmohiv.xls” from the hrp262 website: http://www.stanford.edu/~kcobb/hrp262 to your desktop folder.

Steps: go to the website( right click on “Lab 1 data”( Save target as( Save hmohiv.xls to your desktop.

2. Use point-and-click feature of SAS to assign the library name hrp262 to your desktop folder: 

a. Click the New Library icon on your toolbar (looks like a slamming filing cabinet).
b. In the Name field, type hrp262.
c. Browse to find the path (extension) to your desktop folder. 

d. Click OK to exit the new library screen.

3. Import the hmohiv data using point-and-click*: 

a. Goto: File--> Import Data-->to open Import Wizard

b. Select Microsoft Excel 97, 2000, or 2002 Workbook (default)--> Next-->

c. Browse to find and select the file hmohiv.xls on your desktop. Click Open. Click OK. 

d. Under “what table do you want to import?” leave hmohiv selected-->Next-->

e. Under “Choose the SAS destination” scroll to pick the hrp262 library; then, under member, type: hmohiv to name the dataset hrp262.hmohiv. --->Next-->
f. Browse to find Desktop. Name a file importcode.sas.--->Finish-->
(This last optional step generates the SAS code for importing the hmohiv data, and saves it to a SAS editor file for you—automated programming!)  



*note: import will not work if the original excel dataset is open on your computer.

4. Navigate within your explorer browser to make sure that a SAS dataset hmohiv was created in the hrp262 library.

5. Return to the SAS enhanced editor window. Then goto “File” on your menu bar, and select “Open Program.” Browse to find the program “importcode.sas” on your desktop. Open it.

Should look like:

PROC IMPORT OUT= HRP262.HMOHIV 

            DATAFILE= "C:\Documents and Settings\…your extension…\ hmohiv.xls" 

            DBMS=EXCEL REPLACE;

     SHEET="HMOHIV"; 

     GETNAMES=YES;

     MIXED=NO;

     SCANTEXT=YES;

     USEDATE=YES;

     SCANTIME=YES;

RUN;
Voilà! SAS code is written for you for future reference!  

[Actually, the following code would have been sufficient to create the file…

proc import out=hrp262.hmohiv

datafile= "C:\Documents and Settings\…your extension…\ \hmohiv.xls" 

            dbms=excel replace; 

run;]

6. Use the Explorer Browser to find the SAS dataset hrp262.hmohiv.  Open the dataset in ViewTable Mode. Should contain the variables: id, startdate, enddate, age, drug, censor. 

id: subject’s ID number

startdate: date of entry into study

enddate: date of death or censoring

age: age at entry into study in years

drug: IV drug user (1=yes, 0=no)

censor: 1=died, 0=censored

Make sure to close the dataset when you are done!  (no manipulations of the dataset can be completed if the dataset is open).

7. Fix datetime variables, enddate and startdate and calculate the time that each person was in the study until they died or were censored:

Dealing with date-time variables

/**Dates are automatically imported from excel as date variables (if formatted as date variables in excel)**/
/*Values of date variable represent # of days before or after Jan. 1, 1960**/

/*SAS sees dates as a long number—but you can see dates in any one of the formats given below. Here we ask for the 20April04 format*/
data hrp262.hmohiv;  


set hrp262.hmohiv;


format enddate date.;


format startdate date.;


Time=12*(enddate-startdate)/365.25;  *gives time-to-event in months;

Time=round(time);  *gives rounded month values for ease;
run;

Reference: alternate date formats:

	date.    20April04

date9. 20April2004

day. 20

dowName. Tuesday

dowName3. Tue

monName.  April

monName3. Apr

month. 4
	year2. 04

mmddyy6. 042004

mmddyy8. 04/20/04

mmddyy10. 04/20/2004

weekdate. Tuesday, April 20, 2004

worddate. April 20, 2004

year. 2004




8. Examine the distributions of variables using point-and-click as follows:

a. From the menu select: Solutions(Analysis(Interactive Data Analysis
b. Double click to open: library “hrp262”, dataset “hmohiv”

c. Highlight “censor” variable from the menu select: Analyze(Distribution(Y)
d. From the menu select: Tables(Frequency Counts
e. Scroll down the open analysis window to examine the frequency counts for censor (i.e., how many people died vs. were censored) 

f. Highlight “drug” variable from the menu select: Analyze(Distribution(Y)
g. Place drug and censor distribution windows side by side. Select the green bar that represents censor=1; SAS highlights the values of drug of those with censor=1; allows you to compare the distributions of drug and censor visually!
h. From the menu select: Analyze(Fit (Y X)
i. Use Point-and-click to select age as your X variable and time as your Y-variable. Then click OK to get plot of time*age.

j. Use this feature of SAS to familiarize yourself with any new dataset and to check for outliers, missing data, and values that don’t make sense.

9. Plot survival time against age using PROC GPLOT.  We’ll start with the simplest version and add features as we go along.  Use the following sets of code:

/**Note specification of vertical and horizontal axes scales and use of title statement**/

goptions reset=all;  *resets graphing options; 
proc gplot data=hrp262.hmohiv;

title1 'Time vs. Age: version 1’;


plot time*age /



vaxis = 0 10 20 30 40 50 60  



haxis = 15 20 25 30 35 40 45 50 55 ;

run; quit;   

NOTE: Titles are “global statements”—that means they stay in effect until they are replaced by new ones or removed by entering a blank title: title1  '  ';  
10. Make the graph a little fancier by adding the following features: change the plotting symbol color, shape, and size; reduce minor tick marks to 1 tick between every major tick; and change 'age' label on the x-axis to 'Age (Years)'.  Refer to the GPLOT appendix for more plotting symbol options.  To save time, just add the underlined elements to the previously entered code.

symbol1 value=circle color=red w=2 h=2; *w=width, h=height;

proc gplot data=hrp262.hmohiv;


title1 'Time vs. Age: version 2’;

label age='Age (Years)';


plot time*age /



vaxis = 0 10 20 30 40 50 60 vminor=1 



haxis = 15 20 25 30 35 40 45 50 55 hminor=1;

run; quit;

11. Make the graph even fancier by differentiating between those participants who died and those who were censored (add censor as a classification variable and assign different plotting symbols for censor=1 (died) and censor=0 (censored)).  Also add the following features: use global statements to specify the axes, including turning the y-axis label 90 degrees and changing font size and type; call these axes later within PROC GPLOT.  For symbol 2, try a different plotting symbol using the appendix (I’ve chosen a shamrock).

proc format;

value cens

1="died"
0="censored";

run;


goptions reset=all;

axis1 order= (0 to 60 by 10) 


label=(height= 4pct font='Times New Roman' angle=90); 

axis2 order= (15 to 55 by 10) 

label=(height= 4pct font='Times New Roman'); 

symbol1 v=circle c=blue h=1 w=1;

symbol2 value=% color=red h=1 w=1;


proc gplot data=hrp262.hmohiv;


title1 ‘Fancy Version’;


label time='Survival Time (Months)';


label Age='Age (Years)';


format censor cens.;



plot time*age=censor / 


vaxis = axis1 haxis=axis2 vminor=1 hminor=1;

run; quit;

NOTE: label statements assigned to a variable within a PROC only are valid for duration of that PROC (they are not global statements).

12. Instead of plotting the survival times, we’d like to be able to plot the survival probabilities (i.e., the survival function). It’s not straightforward to make this plot. Luckily, we can just call for a Kaplan-Meier Curve, which gives the empirical survival curve adjusted for censoring:

Non-Parametric Regression in SAS: PROC LIFETEST  

13. Plot the Kaplan-Meier survival curve for the hmohiv data.

/**Kaplan-Meier estimate of survivorship function**/



/*Plot KM curve*/
goptions reset=all;

proc lifetest data=hrp262.hmohiv    plots=(s) graphics censoredsymbol=none;

time time*censor(0); 

title 'Kaplan-Meier plot of survivorship';

symbol v=none ; 

run;

[image: image4.jpg]Kaplan—Meier plot of survivorship





14.  Plot Kaplan-Meier Curves by age, by adding a strata statement in proc lifetest.

proc lifetest data=hrp262.hmohiv plots=(s) graphics censoredsymbol=none;

time time*censor(0); 

strata age(30,40);

symbol v=none ; 

run;

15. It appears that there’s roughly an exponential decrease in survival over time, with rates differing by age group. Let’s use parametric regression to estimate the baseline hazard (and thus baseline survival curve) and the increase in hazard (i.e., decrease in survival) per year of age.  

Parametric Regression in SAS: PROC LIFEREG

Fit an exponential regression model to the data.

proc lifereg data=hrp262.hmohiv;


title 'Exponential curve’;


model time*censor(0)= age /dist=exponential;
run;

16. Examine the output:

                                      Standard   95% Confidence     Chi-

            Parameter     DF Estimate    Error       Limits       Square Pr > ChiSq

            Intercept      1   5.8590   0.5853   4.7119   7.0061  100.22     <.0001

            Age            1  -0.0939   0.0158  -0.1248  -0.0630   35.47     <.0001

            Scale          0   1.0000   0.0000   1.0000   1.0000

            Weibull Shape  0   1.0000   0.0000   1.0000   1.0000

                                 Lagrange Multiplier Statistics


                             Parameter     Chi-Square    Pr > ChiSq

                             Scale             0.0180        0.8932




17.  What does this model look like in terms of the survival curve? Can we plot the resulting survival curve?  Yes, generate a new dataset that contains the predicted survival probabilities (e.g., at 0 months, 1 month, 2 months, etc.) for different ages (e.g., 20 years old at baseline, 30 years old, etc). 



data SurvCurve;

do age=20 to 60 by 10;

do time=0 to 60 by 1;

Hazard=exp(-5.895+.0939*age); 

EstSurv=exp(-Hazard*time);

output;

end;

end;

run;
goptions reset=all;

proc gplot data=SurvCurve;


title1 'Survival Curves by age';


plot EstSurv*time=age;


symbol1 i=join v=none;

run; quit;

APPENDIX A: Some useful logical and mathematical operators and functions:

	Equals: = or eq

Not equal: ^= or ~= or ne

Less then: < or lt, <= or le, 

Greater than: > or gt, >= or ge,
	** power

* multiplication

/ division

+ addition

- subtraction

	INT(v)-returns the integer value (truncates) 

ROUND(v)-rounds a value to the nearest round-off unit 

TRUNC(v)-truncates a numeric value to a specified length

ABS(v)-returns the absolute value 

MOD(v)-calculates the remainder
	SIGN(v)-returns the sign of the argument or 0 

SQRT(v)-calculates the square root 
EXP(v)-raises e (2.71828) to a specified power 
LOG(v)-calculates the natural logarithm (base e) 

LOG10(v)-calculates the common logarithm


APPENDIX B: PROC GPLOT extras
Options for plotting symbols in SAS/GRAPH:

Syntax examples:

symbol1 v=star c=yellow h=1 w=1;

symbol2 value=& color=green h=2 w=2;

Options for plotting lines in SAS/GRAPH:

Syntax examples (place within a symbol statement):  

symbol3 v=none c=black w=2 i=join line=5;

Options for fonts in SAS/GRAPH (Note: non-roman fonts are also available):

Syntax examples (place within a label, title, note, or footnote statement):  


title1 font='SwissXB' 'Figure 1.3, page 16';

Placement of Titles, Labels, Footnotes in SAS/GRAPH

Syntax examples:

title1 'Figure 1.3, page 16';

footnote 'Copyright 2004';

To rotate titles and footnotes:

Syntax examples:


title1 angle=90 'Figure 1.3, page 16';

Symbol1 specifies the first plotting symbol (here only 1 symbol is needed).  


The default plotting symbol is: + (w=1, h=1, black).





This is the formula for hazard rate as specified in our model above. It is a function of age.





PROC FORMAT creates user-defined formats (named whatever you like) that can be called later.


To call formats later:


 format yourvariable yourformatname.;





Global statements can occur outside of PROCS and DATA STEPS.  They stay in effect for all subsequent procedures until you reset them.





Specify distribution of times-to-event to be exponential. 


























This is the estimated survival probability. It is a function of time.





The convention for all survival analyses in SAS is: time*censor(censor value), where time is the time until event or censoring, and censor is a binary indicator variable that tells whether an individual had the event or was censored.  Give SAS the value that represents censored in the parentheses.





Tell sas to omit symbol for each event.  You may also specify this above with  option: “eventsymbol=none”





Requests high resolution graphics





“s” asks for survival plot; see reference page for full list of plotting options





None here eliminates censoring marks.  If you don’t specify, it will give  you annoying circles as the default.





Title, axis, legend, and symbol statements are all global statements.





Title syntax:


Title ‘your title’;





Single quotes


No equals sign.





vaxis (vertical axis) and haxis (horizontal axis)  specifications are options (meaning they are optional commands)


Options appear after a backslash (before the semicolon).





Label syntax:


Label variable= ‘your label’;





Single quotes


Equals sign.





axes syntax:


axisX order=(…units…) label=(…height, font type, and angle of the axis label…);





Translation: 





The Model is:


Ln(HazardRate)= -5.8590 +.0939(age)


(Hazard Rate = � EMBED Equation.3  ���





� EMBED Equation.3  ���





This specifies the entire survival curve for each age:





For example, what is the probability of surviving past 6 months if your age is 0? 


� EMBED Equation.3  ���





For example, what is the probability of surviving past 6 months if your age is 80?


� EMBED Equation.3  ���





      For example, what is the probability of surviving past 30 months if your age is 25?


� EMBED Equation.3  ���








You can also calculate median survival time for each age; for example, for a 25 year old the median survival time is solved as:


� EMBED Equation.3  ���





These are parameters of the weibull distribution, which just equal 1 for an exponential (an exponential is a special case of weibull).





This is testing the null hypothesis that the scale parameter is 1 (and thus that this is actually an exponential). There’s no reason to reject the null, so looks exponential.





The hazard ratio also tells you the relative increase in hazard per year of age.





Hazard Ratio per 1 year increase in age=� EMBED Equation.3  ���=1.098





� EMBED Equation.3  ���





(Note, the hazard ratio is assumed to be constant so it is independent of time)





Translation: there is nearly a 10% increase in the hazard rate (instantaneous mortality rate) for every 1-year increase in age.











Data step:


I’m generating a dataset “SurvCurv” that has the predicted survival probabilities for people aged 20, 30, 40, 50, and 60 (at baseline) across time, (e.g. P(T>0), P(T>1), P(T>2), etc.)


To do this I:


fix the age, starting at 20 (outer do loop)


increment the time from 0 to 60 months (inner do loop)


Calculate the survival probability at each age-time combination


Repeat for different ages





Tells SAS to plot survival probability by time with a separate curve for each age (i.e., stratified by age).





Plot the data! 





Tells SAS to connect the plotting points into a line  (i=join)  for each age and to omit plotting symbols (v=none). 





Note the signs are the opposite from above. A quirk of PROC LIFEREG is that it gives you the negative of the correct estimates.

